
Exercise 1 (Theoretical Foundations): (4 + 4 + 5 = 13 points)

Let ϕ = p(0, 0)∧∀X, Y (p(X, Y )→ p(Y, s(X))) and ψ = ∃Z p(Z, s(Z)) be formulas over the signature
(Σ,∆) with Σ = Σ0 ∪Σ1,Σ0 = {0},Σ1 = {s}, and ∆ = ∆2 = {p}.

a) Prove that {ϕ} |= ψ by means of SLD resolution.

Hint: First transform the formula ϕ ∧ ¬ψ into an equivalent clause set.

b) Explicitly give a Herbrand model of the formula ϕ (i.e., specify a carrier and a meaning for all
function and predicate symbols). You do not have to provide a proof for your answer.

c) Prove or disprove: If K is a set of clauses without variables, S is a model of K, K1, K2 ∈ K and
R is a resolvent of K1 and K2, then S is a model of K ∪ {R}.

Solution:

a)

ϕ ∧ ¬ψ ⇔ p(0, 0) ∧ ∀X, Y (p(X, Y )→ p(Y, s(X))) ∧ ¬∃Z p(Z, s(Z))

⇔ p(0, 0) ∧ ∀X, Y (¬p(X, Y ) ∨ p(Y, s(X))) ∧ ¬∃Z p(Z, s(Z))

⇔ p(0, 0) ∧ ∀X, Y (¬p(X, Y ) ∨ p(Y, s(X))) ∧ ∀Z ¬p(Z, s(Z))

⇔ ∀X, Y, Z (p(0, 0) ∧ (¬p(X, Y ) ∨ p(Y, s(X))) ∧ ¬p(Z, s(Z)))

Thus, the equivalent clause set for ϕ ∧ ¬ψ is {p(0, 0)}, {¬p(X, Y ), p(Y, s(X))}, {¬p(Z, s(Z))}.
We perform SLD resolution on this clause set to show {ϕ} |= ψ.

{p(0, 0)} {¬p(X, Y ), p(Y, s(X))} {¬p(Z, s(Z))}

{¬p(Z,Z)}

Y/Z,X/Z

�

Z/0

Hence, we have proven {ϕ} |= ψ.

�

b) We have S |= ϕ for the Herbrand structure S = (T (Σ), α) with α0 = 0, αs(t) = s(t), and

αp = {(si(0), si(0)) | i ≥ 0} ∪ {(si(0), si+1(0)) | i ≥ 0}

Alternative solution: αp = T (Σ)× T (Σ)

c) Let S be a model of K. Then there is a literal L ∈ K1 such that L ∈ K2 and R = (K1 \ {L}) ∪
(K2 \ {L}). Assume S 6|= K∪ {R}. With S |= K it follows that S 6|= R. If S |= L, S |= K2 implies
S |= K2 \ {L} and hence S |= R. If S |= L, S |= K1 implies S |= K1 \ {L} and hence S |= R.
Therefore, each model of K is also a model of K ∪ {R}.

.

Exercise 2 (Procedural Semantics, SLD tree): (5 + 5 = 10 points)

Consider the following Prolog program P which can be used to check whether a list contains 4 or 6,
but it does not contain any 2 before the first 4 or 6.

e(2).
e(4).
e(6).
p([X|_]):- e(X),!,not(X = 2).
p([_|XS]):- p(XS).
not(X):- X,!,fail.
not(_).

As an example, the query p([1,2,4,8]) would not be provable (since it contains a 2 and there is no 4
or 6 before).

a) The program P ′ results from P by removing both cuts. Consider the following query:

?- p([1,2,4,8]).

For the logic program P ′ (i.e., without the cuts), please show a successful computation for the
query above (i.e., a computation of the form (G,∅) `+P ′ (�, σ) where G = {¬p[1,2,4,8]}). You
may leave out the negations in the queries.

b) Please give a graphical representation of the SLD tree for the query

?- p([1,4]).

in the program P (i.e., with the cuts). For every part of a tree that is cut off by evaluating !,
please indicate the cut by marking the corresponding edge. For the cut-off parts only indicate the
first cut-off goal, but do not evaluate further.

Solution:

a)

`P ′ ({p([1,2,4,8])}, {})
`P ′ ({p([2,4,8])}, {XS/[2,4,8]})
`P ′ ({p([4,8])}, {XS/[2,4,8],XS’/[4,8]})
`P ′ ({e(4),not(4=2)}, {X/4,XS/[2,4,8],XS’/[4,8]})
`P ′ ({not(4=2)}, {X/4,XS/[2,4,8],XS’/[4,8]})
`P ′ (�, {X/4,XS/[2,4,8],XS’/[4,8]})

b) SLD Tree

p([1,4])

e(1),!, not(1=2)

 

p([4])

e(4),!,not(4=2)

!,not(4=2)

not(4=2)

4=2,!,fail

 

�

p([])

//

.

5



Logic Programming SS 2015
Solution - Exam 19.08.2015

Exercise 3 (Fixpoint Semantics): (5 + 3 + 3 = 11 points)

Consider the following logic program P over the signature (Σ,∆) with Σ = {0, s} and ∆ = {p}.
p(0, X).
p(s(X), s(s(Y))) :- p(X, Y).

a) For each n ∈ N explicitly give transnP(∅) in closed form, i.e., using a non-recursive definition.

b) Compute the set lfp(transP).

c) Give F JP, {¬p(s(s(0)), X)}K.

Solution:

Let G be the set of all ground terms, i.e., G = {si(0) | i ∈ N} = T (Σ).

a) foo trans0P(∅) = ∅
trans1P(∅) = {p(0, t) | t ∈ G}
trans2P(∅) = {p(s(0), s2(t)) | t ∈ G} ∪ trans1P(∅)

trans3P(∅) = {p(s2(0), s4(t)) | t ∈ G} ∪ trans2P(∅)

...

transnP(∅) = {p(si(0), s2i(t)) | t ∈ G, 0 ≤ i < n}

b) lfp(transP) = {p(si(0), s2i(t)) | t ∈ G, i ≥ 0} ( = {p(si(0), sj(0)) | i ≥ 0, j ≥ 2i} )

c) F JP, {¬p(s(s(0)), X)}K = {p(s2(0), s4(t)) | t ∈ G} ( = {p(s2(0), s4+i(0)) | i ≥ 0} )

.

Exercise 4 (Definite Logic Programming): (7 + 7 = 14 points)

a) We consider Deterministic Finite Automata (DFAs). An example for such an automaton is given
below. It accepts all words where the number of “a” characters in the word is even.

s0start s1

a

b

a

b

We encode this automaton into Prolog facts as follows:

start(s0).
final(s0).
delta(s0,a,s1).
delta(s1,a,s0).
delta(s1,b,s1).
delta(s0,b,s0).

As a quick reminder: A DFA is a five-tuple (Q,Σ, δ, q0, F ). Here, Q is a set of states (in our
case {s0, s1}), Σ is the set of alphabet symbols (in our case {a, b}). The transition function δ:
Q × Σ 7→ Q maps the current state to the next state given that a certain symbol from Σ was
read. The automaton starts in the start state q0 and accepts the word if it stops in a final state
from the set F ⊆ Q (in our case F = {s0}).
We say that an automaton (Q,Σ, δ, q0, F ) accepts a word w = (a1, a2, . . . , an) ∈ Σn if there is a
run q0

a1−→ q1
a2−→ q2

a3−→ · · · an−→ qn such that for all i ∈ {1 . . . , n} it holds that δ(qi−1, ai) = qi
and qn ∈ F .
In the example above, we encoded the start state q0 with the fact start(s0), the set of final
states F is encoded by the fact final(s0), the transition function is encoded by the delta/3
predicate such that delta(qi, a, qj) holds iff δ(qi , a) = qj . The sets Q and Σ are implicitly
defined by the arguments of delta.

Implement a predicate accepts/1. The query: ?- accepts(Word) should succeed iff the DFA
accepts the given word. In our example, the query ?- accepts([a,b,a]) should succeed but the
query ?- accepts([a,b]) should fail. Your clause for accepts should work for any DFA (i.e.,
for any clauses defining start, final, and delta).

b) Consider the set partition problem: Given a set S = {a1, . . . , an} of integer numbers, find a
partition of S into two sets L and R such that

• Σai∈L ai = Σai∈R ai

• L ∪ R = S

• L ∩ R = ∅.
Implement a predicate partition/3 such that ?- partition(S,L,R) succeeds iff L and R are
a valid partition of S. For example, partition([1,2,3],L,R) should succeed with answer sub-
stitution L = [1,2], R = [3]. On lists with duplicate entries your implementation may behave
arbitrarily.

Solution:

a) Finite Automaton

run_on(State ,[]) :- final(State).
run_on(State ,[C|Word]) :- delta(State ,C,StateN), run_on(StateN ,Word).
accepts(Word):- start(State), run_on(State ,Word).

b) Set Partition

partition_helper ([], [], [], 0, 0).
partition_helper ([X|XS], [X|L], R, SUML , SUMR) :-

partition_helper(XS , L, R, SUMN , SUMR), SUML is SUMN+X.
partition_helper ([X|XS], L, [X|R], SUML , SUMR) :-

partition_helper(XS , L, R, SUML , SUMN), SUMR is SUMN+X.

partition(S, L, R) :- partition_helper(S, L, R, X, X ).

.

Exercise 5 (Meta-Programming): (5 points)

Consider the well known function f ib(i) that returns the i-th Fibonacci number. A possible Prolog
implementation for f ib is:

fib(0,0):- !.
fib(1,1):- !.
fib(X,Y):-

XPP is X-2, fib(XPP ,YPP),
XP is X-1, fib(XP , YP),
Y is YP + YPP ,!.

However, this implementation has roughly exponential runtime. There are many ways to improve this.
In this exercise we will use memoization to improve the performance drastically. Implement a predicate
mem(X) that will try to prove X. If the proof of X succeeds, then the proof of mem(X) is also successful.
Moreover, if the resulting answer substitution σ maps X to a ground term, then σ(X) is added to the
database of program clauses.
Given this predicate, the following definition of fib should only require a linear number of evaluations
of fib.

:-dynamic(fib /2).

fib(0,0):- !.
fib(1,1):- !.
fib(X,Y):-

XPP is X-2, mem(fib(XPP ,YPP)),
XP is X-1, mem(fib(XP , YP)),
Y is YP + YPP ,!.

Hints:

• You may use the built-in predicates assertz/1, asserta/1, ground/1 and the cut operator.

Solution:

mem(X):- X, mem_if_ground(X).

mem_if_ground(X):- ground(X),asserta(X).
mem_if_ground(_).

.

Exercise 6 (Difference Lists): (7 points)

Consider the following logic program P.
append_diff(A-B,B-C,A-C).

r([1,2,3,4|X]-X).
r(Z):- r([H|T]-X), append_diff(T-X,[H|Y]-Y,Z).

Explicitly give the set of all answer substitutions for the query ?- r(ZS) (up to variable renaming).

Solution:

ZS = [1,2,3,4|X]-X,
ZS = [2,3,4,1|X]-X,
ZS = [3,4,1,2|X]-X,
ZS = [4,1,2,3|X]-X

.

